
Úvod 

8 přednáška Pole prvků - karty

7

Náhodná čísla (cisla/nahcisla1.frm)

Naším úkolem je vygenerovat řadu tlačítek, jejichž nápisy tvoří náhodně nagenerovaná čísla.

Generátor náhodných čísel:

Příkaz Randomize způsobí, že náhodně vygenerovaná čísla jsou při každém spuštění programu jiná. Pokud bych tento příkaz na inicializaci generátoru náhodných čísel nepoužila, generovala by se při každém běhu programu čísla stejná.

Ke generování náhodných čísel použijeme funkci Rnd (v Excelu funkce NÁHČÍSLO), pokud budeme chtít získávat čísla celá doplníme funkcí Int. Tato funkce odtrhne desetinou část čísla a vrátí celočíselnou hodnotu typu Integer.

Generování celého čísla z intervalu <1;10>:

Cisla(i).Caption = Str(Int(Rnd() * 10 + 1)) 'nastavení čísla

Po stisku tlačítka Generuj se vytvoří 10 tlačítek, s náhodně vygenerovanými čísly, po stisku tlačítka Smaž se tlačítka odstraní z formuláře.

Private Sub cbGeneruj_Click()

Dim i As Integer 'řidicí proměnná cyklu

Dim x As Integer 'pozice tlačítka

Dim y As Integer 'pozice tlačítka

x = Cisla(0).Left   'podle vzoru počáteční pozice

y = Cisla(0).Top

' generuji řadu náhodných čísel

' Varianta1 konkrétní počet 10

Randomize  'znáhodní generátor náhodných čísel

For i = 1 To 10

    Load Cisla(i) ' nový prvek Label

    Cisla(i).Left = x 'umístění

    Cisla(i).Top = y

    Cisla(i).Visible = True

    Cisla(i).Caption = Str(Int(Rnd() * 10 + 1)) 'nastavení čísla

    x = x + Cisla(0).Width + Mezera 'příprava pozice

Next i

cbGeneruj.Visible = False

cbSmaz.Visible = True

End Sub

Upozornění na opakující se čísla (cisla/nahcisla2.frm)
Teď tento příklad maličko ztížíme. Budeme při generování čísel dávat pozor na to, zda se vygenerovaná hodnota čísla již nevyskytla a pokud ano, text tohoto tlačítka zobrazíme červeně.

Jak budeme tuto kontrolu provádět? Vždy, když přidám nové tlačítko a vygeneruji si jeho hodnotu, musím projít tlačítka předchozí (s indexy 1 – i-1, pokud i je aktuální index) opět pomocí For cyklu a hledat, zda se tam tato hodnota již nevyskytla. Pokud takovou hodnotu naleznu, změním barvu nápisu aktuálního tlačítka a hledání mohu ukončit (předčasné ukončení cyklu Exit For.

Private Sub cbGeneruj_Click()

 . . . 
For i = 1 To 10

    . . .

    Cisla(i).Caption = Str(Int(Rnd() * 10 + 1)) 'nastavení čísla

    'prohledavani predchozich tlacitek

    For j = 1 To i - 1

        If Cisla(j).Caption = Cisla(i).Caption Then
            Cisla(i).ForeColor = vbRed

            Exit For 'predcasne ukonceni cyklu

        End If
    Next j

    x = x + Cisla(0).Width + Mezera 'příprava pozice

Next i

. . .

End Sub

 Karty (karty/karty0.frm)

Na formuláři je jako vzor umístěno 10 různých obrázků, z těchto obrázků potřebuji vytvářet variace s opakováním páté třídy. Tedy chci vybrat 5 „karet“ a zobrazit je za sebou. Vzory obrázků tvoří pole prvků typu Image vzor(1 To 10), toto pole bylo vytvořeno  v režimu návrhu. Pozadí obrázků představuje prvek opět typu Image pozadi.

Vybrané obrázky budou umístěny do pole prvků karta(0 to 5), kde prvek s indexem 0 je vzorem budoucího pole a jako jediný je na formulář umístěn již v režimu návrhu. Ostatní prvky tohoto pole vzniknou až za běhu programu v událostní proceduře Form_Load.

Zobrazováni karet (karty/karty1.frm)
Aplikace je ovládána pomocí tlačítka Otoč. Které střídavě otáčí pozadí karet a náhodně vygenerované obrázky karet.

Příprava karet

Private Sub Form_Load()
Dim i As Integer 'pocitadlo For cyklu
Dim x As Integer 'pozice karty
Dim y As Integer 'pozice karty

x = karta(0).Left   'podle vzoru počáteční pozice
y = karta(0).Top

Rub = True 'karty jsou otoceny rubovou stranou

Randomize

'pripravim 5 tlacitek, opatrenych pozadim
For i = 1 To N
    Load karta(i) ' nový prvek Label
    karta(i).Left = x 'umístění
    karta(i).Top = y
    karta(i).Visible = True
    karta(i).Picture = pozadi.Picture
    x = x + karta(0).Width + Mezera 'příprava pozice
Next i
'zadna karta dosud nebyla vybrana

End Sub
Reakce na stisknutí tlačítka Otoc.

Private Sub cbOtoc_Click()
Dim i As Integer 'pocitadlo For cyklu
Dim indexk As Integer   'index vybrane karty


If Rub Then
    'vybiram a zobrazuji pet karet v nahodnem poradi

    For i = 1 To N
        indexk = Int(Rnd() * 10 + 1)
        karta(i) = vzor(indexk).Picture
    Next i
Else
    'otocima zpet na rubovou stranu
    For i = 1 To N
        karta(i) = pozadi.Picture
    Next i

End If
'karty jsme obratili
Rub = Not (Rub)
End Sub
Příště si ukážeme jak zobrazovat karty tak, aby se navzájem neopakovaly.

Permutace bez opakování (cisla/nahcisla3.frm)
V okamžiku, kdy zjistíme, že námi nově generované číslo již bylo jednou použito musíme celý proces generování náhodného čísla a jeho kontroly opakovat. Toto opakované generování probíhá tak dlouho dokud se nám nepodaří vygenerovat takovou hodnotu, která se dosud nevyskytla.

Bude se tedy jednat o cyklus, cyklus řízený podmínkou. Použijeme cyklus 

Do 

příkazy

Loop Until podmínka

Příkazy uzavřené mezi klíčová slova Do  a Loop se budou opakovat tak dlouho, dokud nebude platit podmínka. V okamžiku splnění podmínky se pokračuje na první řádce za tělem cyklu.

Kontrolu generování můžeme realizovat pomocí boolovské proměnné neopak.

Private Sub cbGeneruj_Click()

Dim neopak As Boolean 'kontrola zda se èísla neopakují

Randomize  'znáhodní generátor náhodných èísel

For i = 1 To 10

    Do

    'hodnotu generuji tak dlouho, dokud se mi nepodari vygenerovat promennou, ktera

    'se dosud nevyskytla

    neopak = True 'hodnota se neopakuje

    Cisla(i).Caption = Str(Int(Rnd() * 10 + 1)) 'nastavení èísla

    'prohledavani predchozich tlacitek

    For j = 1 To i - 1

        If Cisla(j).Caption = Cisla(i).Caption Then

            neopak = False 'hodnota se opakuje

            Exit For 'predcasne ukonceni cyklu

        End If

    Next j

    Loop Until neopak

    x = x + Cisla(0).Width + Mezera 'pøíprava pozice

Next i

..

End Sub
Variace (permutace) bez opakování

Nyní si naprogramujeme opět generování kartiček, ale tentokrát efektivně. Nebudeme již opakovat náhodný výběr kartičky a test, zda kartička byla nebo nebyla použita. Chceme v každém kroku vybrat náhodně právě jednu kartičku.

Konkrétní příklad:

Můžeme vybírat z deseti karet. Poprvé budu opravdu losovat z těchto deseti karet, a to číslo, které padne, použiji. Všechny kartičky jsou volné, tudíž proměnná pom v mém programu roste stejně rychle jako proměnná indexk.

Při druhém výběru již vybírám náhodně pouze z množiny devíti volných kartiček. Opět náhodně zvolím, kolikátou z volných chci vybrat. Toto náhodně vygenerované číslo, však již neříká nic o tom, jaký je index té vybírané kartičky. Musím procházet volné kartičky a počítat je. Až se dostanu k hledané volné kartičce.

Private Sub cbOtoc_Click()
Dim i As Integer 'pocitadlo For cyklu
Dim index_v As Integer   'index vybrane karty
Dim kol_vol As Integer   'kolikatou volnou kartu vybrat
Dim pom As Integer 'pocitam volne karty
Dim Zbyva As Integer 'kolik zbyva nepouzitych karticek


If Rub Then
    'vybiram a zobrazuji pet karet v nahodnem poradi
    Zbyva = 10
    For i = 1 To N
    'chci obsadit N karet
        kol_vol = Int(Rnd() * Zbyva + 1)
        'vybiram si z dosud Zbyva nepouzitych vzoru
        index_v = 0 'prochazim vzory
        pom = 1
        'spocitam si volne karty a najdu kol_vol kartu
        'pom slouzi k spocteni volnych karet
        Do While pom <= kol_vol
         index_v = index_v + 1 'karta kterou testuji
         If Not (vzor_obsazen(index_v)) Then
            pom = pom + 1
         End If

        Loop

        karta(i) = vzor(index_v).Picture 'zobrazim kartu
        vzor_obsazen(index_v) = True 'oznacim ji jako pouzitou
        Zbyva = Zbyva - 1 'priste vybiram z mensi mnoziny
    Next i
Else
    'otocima zpet na rubovou stranu
    For i = 1 To N
        karta(i) = pozadi.Picture
    Next i
    'nastavim ze vsechny karty jsou volne pomoci vlastnosti Tag
    For i = 1 To 10
        vzor_obsazen(i) = False
    Next i
End If
'karty jsme obratili
Rub = Not (Rub)
End Sub
 Abeceda (Abeceda/Abeceda.vbp)
Tento příklad je ukázkou toho, že mohu jako index použít například písmeno. Mým úkolem je na formulář vygenerovat tlačítka anglické abecedy.

Opět použiji jedno neviditelné tlačítko jako vzor budoucího pole pomocí kterého nageneruji celou abecedu.

Každému písmenu je přiřazen kód v Ascii tabulce – číslo 0-255 ASCI, pomocí funkce Asc mohu převést písmeno na číslo a např. pomocí něj indexovat. Využíváme zde vlastnosti, že velká písmena tvoří souvislou řadu

For i = Asc("A") To Asc("Z")

Next i

Funkce Chr() naopak vytvoří z čísla písmeno.

Abeceda(i).Caption = Chr(i)

V současné době se používá kódování znaků pomocí Unicode (2 byty) – funkce AscW(), ChrW().Znaky 0-31 tzv. netisknutelné – řídicí.

Reakce na stisk klávesy (Abeceda/AbecedaK.vbp)

Zatím jsme písmena vybírali klepnutím myši na příslušné tlačítko, ale někdy potřebujeme reagovat i na stisk klávesy na klávesnici.

Private Sub Form_KeyPress(KeyAscii As Integer)

    'chci reagovat na stisknuté písmeno na klávesnici jeho zobrazením

    'otestuji zda bylo stisknuto velké písmeno

    If KeyAscii >= Asc("A") And KeyAscii <= Asc("Z") Then

        Pismeno.Caption = Abeceda(KeyAscii).Caption

    End If

    'program by take mel reagovat na stisk maleho pismena

    If KeyAscii >= Asc("a") And KeyAscii <= Asc("z") Then

    'pozor malé písmeno musím pøevést na velké – funkce UCase

        Pismeno.Caption = Abeceda(Asc(UCase(Chr(KeyAscii)))).Caption

    End If

End Sub

Událostní procedura Form_KeyPress slouží jen k detekci těch kláves, které mají za sebou odpovídající znak – nepoužívá se k detekci stisku např. kurzorových šipek. Zde použijte událostní procedury Form_KeyDown nebo Form_KeyUp.

Práce s textovými řetězci (Abeceda/Text1.vbp)
Pro ilustraci funkcí pro práci s textem uděláme jednoduchý příklad, kde máme v poli řetězců různá hesla (tajenky). Z těchto hesel jedno vybereme a předložíme ho k hádání uživateli. V praxi by to asi fungovalo tak, že hesla by byly uloženy v souborech  ze kterých by je program četl (letos se nebudeme učit). Další možností je uložit hesla do ListBoxu, ale pro naše účely zcela postačí pole

Náhodné vybrání hesla a jeho zobrazení do prvku typu Label, bychom měli již bez problémů zvládnout.

Zobrazení textu po písmenech (Abeceda/Text2.vbp)
Pro programování různých her a doplňovaček budeme potřebovat umět rozložit textový řetězec do jednotlivých znaků.. Pro reprezentaci jednotlivých znaků můžeme opět použít pole řetězců, mohli bychom ale znaky také ukládat v celočíselném poli pomocí jejich ASCII kódů.

Použijeme „dynamické pole“ dopředu nevíme, kolik znaků slovo má.

Dim slovo() As String

Jednoduchý příklad nám ukáže jak heslo připravit do jednotlivých políček typu Label. Náhodně vybereme heslo a podle jeho délky připravíme příslušný počet políček. Heslo rozdělíme do pomocného textového pole na jednotlivá písmena.

Index = Int(Rnd() * 10 + 1)

delka = Len(heslo(Index))

pomslovo = UCase(heslo(Index)) 'prevedeni vybraneho slova do velkych pismen

'rozdeleni slova na pismena

ReDim slovo(1 To delka)

For i = 1 To delka

 slovo(i) = Mid(pomslovo, i, 1)

Next i

V poli slovo máme nyní heslo rozložené po jednotlivých písmenech.

Po klepnutí na políčko se nám zobrazí písmeno, které jsme uložili do pomocného pole.

Private Sub Pismeno_Click(Index As Integer)

'zobrazi se pismeno, ktere patri na danou pozici

Pismeno(Index).Caption = slovo(Index)

End Sub

